/**
 *Submitted for verification at polygonscan.com on 2023-04-26
*/

// SPDX-License-Identifier: MIT
// File: @openzeppelin/contracts/utils/math/Math.sol


// OpenZeppelin Contracts (last updated v4.8.0) (utils/math/Math.sol)

pragma solidity ^0.8.0;

/**
 * @dev Standard math utilities missing in the Solidity language.
 */
library Math {
    enum Rounding {
        Down, // Toward negative infinity
        Up, // Toward infinity
        Zero // Toward zero
    }

    /**
     * @dev Returns the largest of two numbers.
     */
    function max(uint256 a, uint256 b) internal pure returns (uint256) {
        return a > b ? a : b;
    }

    /**
     * @dev Returns the smallest of two numbers.
     */
    function min(uint256 a, uint256 b) internal pure returns (uint256) {
        return a < b ? a : b;
    }

    /**
     * @dev Returns the average of two numbers. The result is rounded towards
     * zero.
     */
    function average(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b) / 2 can overflow.
        return (a & b) + (a ^ b) / 2;
    }

    /**
     * @dev Returns the ceiling of the division of two numbers.
     *
     * This differs from standard division with `/` in that it rounds up instead
     * of rounding down.
     */
    function ceilDiv(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b - 1) / b can overflow on addition, so we distribute.
        return a == 0 ? 0 : (a - 1) / b + 1;
    }

    /**
     * @notice Calculates floor(x * y / denominator) with full precision. Throws if result overflows a uint256 or denominator == 0
     * @dev Original credit to Remco Bloemen under MIT license (https://xn--2-umb.com/21/muldiv)
     * with further edits by Uniswap Labs also under MIT license.
     */
    function mulDiv(
        uint256 x,
        uint256 y,
        uint256 denominator
    ) internal pure returns (uint256 result) {
        unchecked {
            // 512-bit multiply [prod1 prod0] = x * y. Compute the product mod 2^256 and mod 2^256 - 1, then use
            // use the Chinese Remainder Theorem to reconstruct the 512 bit result. The result is stored in two 256
            // variables such that product = prod1 * 2^256 + prod0.
            uint256 prod0; // Least significant 256 bits of the product
            uint256 prod1; // Most significant 256 bits of the product
            assembly {
                let mm := mulmod(x, y, not(0))
                prod0 := mul(x, y)
                prod1 := sub(sub(mm, prod0), lt(mm, prod0))
            }

            // Handle non-overflow cases, 256 by 256 division.
            if (prod1 == 0) {
                return prod0 / denominator;
            }

            // Make sure the result is less than 2^256. Also prevents denominator == 0.
            require(denominator > prod1);

            ///////////////////////////////////////////////
            // 512 by 256 division.
            ///////////////////////////////////////////////

            // Make division exact by subtracting the remainder from [prod1 prod0].
            uint256 remainder;
            assembly {
                // Compute remainder using mulmod.
                remainder := mulmod(x, y, denominator)

                // Subtract 256 bit number from 512 bit number.
                prod1 := sub(prod1, gt(remainder, prod0))
                prod0 := sub(prod0, remainder)
            }

            // Factor powers of two out of denominator and compute largest power of two divisor of denominator. Always >= 1.
            // See https://cs.stackexchange.com/q/138556/92363.

            // Does not overflow because the denominator cannot be zero at this stage in the function.
            uint256 twos = denominator & (~denominator + 1);
            assembly {
                // Divide denominator by twos.
                denominator := div(denominator, twos)

                // Divide [prod1 prod0] by twos.
                prod0 := div(prod0, twos)

                // Flip twos such that it is 2^256 / twos. If twos is zero, then it becomes one.
                twos := add(div(sub(0, twos), twos), 1)
            }

            // Shift in bits from prod1 into prod0.
            prod0 |= prod1 * twos;

            // Invert denominator mod 2^256. Now that denominator is an odd number, it has an inverse modulo 2^256 such
            // that denominator * inv = 1 mod 2^256. Compute the inverse by starting with a seed that is correct for
            // four bits. That is, denominator * inv = 1 mod 2^4.
            uint256 inverse = (3 * denominator) ^ 2;

            // Use the Newton-Raphson iteration to improve the precision. Thanks to Hensel's lifting lemma, this also works
            // in modular arithmetic, doubling the correct bits in each step.
            inverse *= 2 - denominator * inverse; // inverse mod 2^8
            inverse *= 2 - denominator * inverse; // inverse mod 2^16
            inverse *= 2 - denominator * inverse; // inverse mod 2^32
            inverse *= 2 - denominator * inverse; // inverse mod 2^64
            inverse *= 2 - denominator * inverse; // inverse mod 2^128
            inverse *= 2 - denominator * inverse; // inverse mod 2^256

            // Because the division is now exact we can divide by multiplying with the modular inverse of denominator.
            // This will give us the correct result modulo 2^256. Since the preconditions guarantee that the outcome is
            // less than 2^256, this is the final result. We don't need to compute the high bits of the result and prod1
            // is no longer required.
            result = prod0 * inverse;
            return result;
        }
    }

    /**
     * @notice Calculates x * y / denominator with full precision, following the selected rounding direction.
     */
    function mulDiv(
        uint256 x,
        uint256 y,
        uint256 denominator,
        Rounding rounding
    ) internal pure returns (uint256) {
        uint256 result = mulDiv(x, y, denominator);
        if (rounding == Rounding.Up && mulmod(x, y, denominator) > 0) {
            result += 1;
        }
        return result;
    }

    /**
     * @dev Returns the square root of a number. If the number is not a perfect square, the value is rounded down.
     *
     * Inspired by Henry S. Warren, Jr.'s "Hacker's Delight" (Chapter 11).
     */
    function sqrt(uint256 a) internal pure returns (uint256) {
        if (a == 0) {
            return 0;
        }

        // For our first guess, we get the biggest power of 2 which is smaller than the square root of the target.
        //
        // We know that the "msb" (most significant bit) of our target number `a` is a power of 2 such that we have
        // `msb(a) <= a < 2*msb(a)`. This value can be written `msb(a)=2**k` with `k=log2(a)`.
        //
        // This can be rewritten `2**log2(a) <= a < 2**(log2(a) + 1)`
        // → `sqrt(2**k) <= sqrt(a) < sqrt(2**(k+1))`
        // → `2**(k/2) <= sqrt(a) < 2**((k+1)/2) <= 2**(k/2 + 1)`
        //
        // Consequently, `2**(log2(a) / 2)` is a good first approximation of `sqrt(a)` with at least 1 correct bit.
        uint256 result = 1 << (log2(a) >> 1);

        // At this point `result` is an estimation with one bit of precision. We know the true value is a uint128,
        // since it is the square root of a uint256. Newton's method converges quadratically (precision doubles at
        // every iteration). We thus need at most 7 iteration to turn our partial result with one bit of precision
        // into the expected uint128 result.
        unchecked {
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            return min(result, a / result);
        }
    }

    /**
     * @notice Calculates sqrt(a), following the selected rounding direction.
     */
    function sqrt(uint256 a, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = sqrt(a);
            return result + (rounding == Rounding.Up && result * result < a ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 2, rounded down, of a positive value.
     * Returns 0 if given 0.
     */
    function log2(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >> 128 > 0) {
                value >>= 128;
                result += 128;
            }
            if (value >> 64 > 0) {
                value >>= 64;
                result += 64;
            }
            if (value >> 32 > 0) {
                value >>= 32;
                result += 32;
            }
            if (value >> 16 > 0) {
                value >>= 16;
                result += 16;
            }
            if (value >> 8 > 0) {
                value >>= 8;
                result += 8;
            }
            if (value >> 4 > 0) {
                value >>= 4;
                result += 4;
            }
            if (value >> 2 > 0) {
                value >>= 2;
                result += 2;
            }
            if (value >> 1 > 0) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 2, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log2(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log2(value);
            return result + (rounding == Rounding.Up && 1 << result < value ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 10, rounded down, of a positive value.
     * Returns 0 if given 0.
     */
    function log10(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >= 10**64) {
                value /= 10**64;
                result += 64;
            }
            if (value >= 10**32) {
                value /= 10**32;
                result += 32;
            }
            if (value >= 10**16) {
                value /= 10**16;
                result += 16;
            }
            if (value >= 10**8) {
                value /= 10**8;
                result += 8;
            }
            if (value >= 10**4) {
                value /= 10**4;
                result += 4;
            }
            if (value >= 10**2) {
                value /= 10**2;
                result += 2;
            }
            if (value >= 10**1) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 10, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log10(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log10(value);
            return result + (rounding == Rounding.Up && 10**result < value ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 256, rounded down, of a positive value.
     * Returns 0 if given 0.
     *
     * Adding one to the result gives the number of pairs of hex symbols needed to represent `value` as a hex string.
     */
    function log256(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >> 128 > 0) {
                value >>= 128;
                result += 16;
            }
            if (value >> 64 > 0) {
                value >>= 64;
                result += 8;
            }
            if (value >> 32 > 0) {
                value >>= 32;
                result += 4;
            }
            if (value >> 16 > 0) {
                value >>= 16;
                result += 2;
            }
            if (value >> 8 > 0) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 10, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log256(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log256(value);
            return result + (rounding == Rounding.Up && 1 << (result * 8) < value ? 1 : 0);
        }
    }
}

// File: @openzeppelin/contracts/utils/Strings.sol


// OpenZeppelin Contracts (last updated v4.8.0) (utils/Strings.sol)

pragma solidity ^0.8.0;


/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _SYMBOLS = "0123456789abcdef";
    uint8 private constant _ADDRESS_LENGTH = 20;

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        unchecked {
            uint256 length = Math.log10(value) + 1;
            string memory buffer = new string(length);
            uint256 ptr;
            /// @solidity memory-safe-assembly
            assembly {
                ptr := add(buffer, add(32, length))
            }
            while (true) {
                ptr--;
                /// @solidity memory-safe-assembly
                assembly {
                    mstore8(ptr, byte(mod(value, 10), _SYMBOLS))
                }
                value /= 10;
                if (value == 0) break;
            }
            return buffer;
        }
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        unchecked {
            return toHexString(value, Math.log256(value) + 1);
        }
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }

    /**
     * @dev Converts an `address` with fixed length of 20 bytes to its not checksummed ASCII `string` hexadecimal representation.
     */
    function toHexString(address addr) internal pure returns (string memory) {
        return toHexString(uint256(uint160(addr)), _ADDRESS_LENGTH);
    }
}

// File: @openzeppelin/contracts/utils/Base64.sol


// OpenZeppelin Contracts (last updated v4.7.0) (utils/Base64.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides a set of functions to operate with Base64 strings.
 *
 * _Available since v4.5._
 */
library Base64 {
    /**
     * @dev Base64 Encoding/Decoding Table
     */
    string internal constant _TABLE = "ABCDEFGHIJKLMNOPQRSTUVWXYZabcdefghijklmnopqrstuvwxyz0123456789+/";

    /**
     * @dev Converts a `bytes` to its Bytes64 `string` representation.
     */
    function encode(bytes memory data) internal pure returns (string memory) {
        /**
         * Inspired by Brecht Devos (Brechtpd) implementation - MIT licence
         * https://github.com/Brechtpd/base64/blob/e78d9fd951e7b0977ddca77d92dc85183770daf4/base64.sol
         */
        if (data.length == 0) return "";

        // Loads the table into memory
        string memory table = _TABLE;

        // Encoding takes 3 bytes chunks of binary data from `bytes` data parameter
        // and split into 4 numbers of 6 bits.
        // The final Base64 length should be `bytes` data length multiplied by 4/3 rounded up
        // - `data.length + 2`  -> Round up
        // - `/ 3`              -> Number of 3-bytes chunks
        // - `4 *`              -> 4 characters for each chunk
        string memory result = new string(4 * ((data.length + 2) / 3));

        /// @solidity memory-safe-assembly
        assembly {
            // Prepare the lookup table (skip the first "length" byte)
            let tablePtr := add(table, 1)

            // Prepare result pointer, jump over length
            let resultPtr := add(result, 32)

            // Run over the input, 3 bytes at a time
            for {
                let dataPtr := data
                let endPtr := add(data, mload(data))
            } lt(dataPtr, endPtr) {

            } {
                // Advance 3 bytes
                dataPtr := add(dataPtr, 3)
                let input := mload(dataPtr)

                // To write each character, shift the 3 bytes (18 bits) chunk
                // 4 times in blocks of 6 bits for each character (18, 12, 6, 0)
                // and apply logical AND with 0x3F which is the number of
                // the previous character in the ASCII table prior to the Base64 Table
                // The result is then added to the table to get the character to write,
                // and finally write it in the result pointer but with a left shift
                // of 256 (1 byte) - 8 (1 ASCII char) = 248 bits

                mstore8(resultPtr, mload(add(tablePtr, and(shr(18, input), 0x3F))))
                resultPtr := add(resultPtr, 1) // Advance

                mstore8(resultPtr, mload(add(tablePtr, and(shr(12, input), 0x3F))))
                resultPtr := add(resultPtr, 1) // Advance

                mstore8(resultPtr, mload(add(tablePtr, and(shr(6, input), 0x3F))))
                resultPtr := add(resultPtr, 1) // Advance

                mstore8(resultPtr, mload(add(tablePtr, and(input, 0x3F))))
                resultPtr := add(resultPtr, 1) // Advance
            }

            // When data `bytes` is not exactly 3 bytes long
            // it is padded with `=` characters at the end
            switch mod(mload(data), 3)
            case 1 {
                mstore8(sub(resultPtr, 1), 0x3d)
                mstore8(sub(resultPtr, 2), 0x3d)
            }
            case 2 {
                mstore8(sub(resultPtr, 1), 0x3d)
            }
        }

        return result;
    }
}

// File: @openzeppelin/contracts/utils/Context.sol


// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/access/Ownable.sol


// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;


/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: Forge Token/Forge_Metadata.sol

// Contract based on https://docs.openzeppelin.com/contracts/4.x/erc721

pragma solidity ^0.8.4;





interface ForgeToken {
    struct TokenAttributes { uint16 xp; bool isSpecialVariant; }
    
    function tokenAttributes(uint _tokenId)
        external
        view
        returns(TokenAttributes memory);
}

//=============================================================================
/// @title Forge_Metadata
/// @dev Implementation of the tokenURI function.
///  - This implementation was created to support returning metadata as data
///  to remove the requirement of hosting a JSON object on a web2 endpoint,
///  and to make metadata immediately queryable upon mint.
///  - This is implemented as an independent smart contract to enable future
///  modification of returned information.
//=============================================================================
contract Forge_Metadata is Ownable {

    // Token Contract for attributes reference
    ForgeToken public forgeTokenContract;
    // base URI for images/mp4
    string public baseURI = "https://forge.gg/nft/images/";
    string public defaultURI = "hot";
    string public specialVariantURI = "cold";
    // extensions for images/mp4
    string public imageExtension = ".png";
    string public animationExtension = ".mp4";
    // token names
    string public defaultName = "Forge Day 0 Badge";
    string public specialVariantName = "Forge Day 0 Cold Fire Badge";
    // token trait name and values
    string public xpTraitName = "XP";
    string public mintNumberTraitName = "Limited Edition Number";
    string public specialVariantTraitName = "Heat";
    string public defaultTraitValue = "Fire";
    string public specialVariantTraitValue = "Cold Fire";
    // token description
    string public defaultDescription = "The Forge Day 0 badge is a symbol of your early commitment to the Forge community and unlocks unique perks on the Forge platform.";
    string public specialVariantDescription = "The Forge Day 0 badge is a symbol of your early commitment to the Forge community and unlocks unique perks on the Forge platform.";
    // display types
    string public xpDisplayType = "boost_number";
    string public mintNumberDisplayType = "number";

    //-------------------------------------------------------------------------
    //-------------------------------------------------------------------------
    function setContract_ForgeToken(address _forgeTokenAddress)
        external
        onlyOwner
    {
        forgeTokenContract = ForgeToken(_forgeTokenAddress);
    }

    //-------------------------------------------------------------------------
    //-------------------------------------------------------------------------
    function setURI_Base(string calldata _baseURI)
        external
        onlyOwner
    {
        baseURI = _baseURI;
    }

    //-------------------------------------------------------------------------
    //-------------------------------------------------------------------------
    function setURI_Default(string calldata _defaultURI)
        external
        onlyOwner
    {
        defaultURI = _defaultURI;
    }

    //-------------------------------------------------------------------------
    //-------------------------------------------------------------------------
    function setURI_SpecialVariant(string calldata _specialVariantURI)
        external
        onlyOwner
    {
        specialVariantURI = _specialVariantURI;
    }

    //-------------------------------------------------------------------------
    //-------------------------------------------------------------------------
    function setImageExtension(string calldata _extension)
        external
        onlyOwner
    {
        imageExtension = _extension;
    }

    //-------------------------------------------------------------------------
    //-------------------------------------------------------------------------
    function setAnimationExtension(string calldata _extension)
        external
        onlyOwner
    {
        animationExtension = _extension;
    }

    //-------------------------------------------------------------------------
    //-------------------------------------------------------------------------
    function setName_Default(string calldata _defaultName)
        external
        onlyOwner
    {
        defaultName = _defaultName;
    }

    //-------------------------------------------------------------------------
    //-------------------------------------------------------------------------
    function setName_SpecialVariant(string calldata _specialVariantName)
        external
        onlyOwner
    {
        specialVariantName = _specialVariantName;
    }

    //-------------------------------------------------------------------------
    //-------------------------------------------------------------------------
    function setTraitName_XP(string calldata _xpTraitName)
        external
        onlyOwner
    {
        xpTraitName = _xpTraitName;
    }

    //-------------------------------------------------------------------------
    //-------------------------------------------------------------------------
    function setTraitName_MintNumber(string calldata _mintNumberTraitName)
        external
        onlyOwner
    {
        mintNumberTraitName = _mintNumberTraitName;
    }

    //-------------------------------------------------------------------------
    //-------------------------------------------------------------------------
    function setTraitName_SpecialVariant(string calldata _specialVariantTraitName)
        external
        onlyOwner
    {
        specialVariantTraitName = _specialVariantTraitName;
    }

    //-------------------------------------------------------------------------
    //-------------------------------------------------------------------------
    function setTraitValue_Default(string calldata _defaultTraitValue)
        external
        onlyOwner
    {
        defaultTraitValue = _defaultTraitValue;
    }

    //-------------------------------------------------------------------------
    //-------------------------------------------------------------------------
    function setTraitValue_SpecialVariant(string calldata _specialVariantTraitValue)
        external
        onlyOwner
    {
        specialVariantTraitValue = _specialVariantTraitValue;
    }

    //-------------------------------------------------------------------------
    //-------------------------------------------------------------------------
    function setDescription_Default(string calldata _defaultDescription)
        external
        onlyOwner
    {
        defaultDescription = _defaultDescription;
    }

    //-------------------------------------------------------------------------
    //-------------------------------------------------------------------------
    function setDescription_SpecialVariant(string calldata _specialVariantDescription)
        external
        onlyOwner
    {
        specialVariantDescription = _specialVariantDescription;
    }

    //-------------------------------------------------------------------------
    //-------------------------------------------------------------------------
    function setDisplayType_XP(string calldata _xpDisplayType)
        external
        onlyOwner
    {
        xpDisplayType = _xpDisplayType;
    }

    //-------------------------------------------------------------------------
    //-------------------------------------------------------------------------
    function setDisplayType_MintNumber(string calldata _mintNumberDisplayType)
        external
        onlyOwner
    {
        mintNumberDisplayType = _mintNumberDisplayType;
    }

    //-------------------------------------------------------------------------
    /// @notice Generates a URL containing metadata as data
    /// @dev This is an extremely costly function. Never call this as part of
    ///  a transaction.
    //-------------------------------------------------------------------------
    function tokenURI(uint _tokenId) public view returns (string memory) {
        ForgeToken.TokenAttributes memory tokenData = forgeTokenContract.tokenAttributes(_tokenId);

        bytes memory dataURI = generateMetadata(tokenData, _tokenId);

        return string(
            abi.encodePacked(
                "data:application/json;base64,",
                Base64.encode(dataURI)
            )
        );
    }

    function generateMetadata(ForgeToken.TokenAttributes memory tokenData, uint _tokenId) internal view returns (bytes memory) {
        string memory tokenId = Strings.toString(_tokenId);
        string memory imgUrl = generateImgUrl(tokenData.isSpecialVariant);
        string memory animationURL = generateAnimationUrl(tokenData.isSpecialVariant);
        string memory attributes = generateAttributes(tokenData, tokenId);

        return abi.encodePacked(
            "{\n",
            "\t\"name\": \"", tokenData.isSpecialVariant ? specialVariantName : defaultName, " #", tokenId, "\",\n",
            "\t\"description\": \"", tokenData.isSpecialVariant ? specialVariantDescription : defaultDescription, "\",\n",
            "\t\"image\": \"", imgUrl, "\",\n",
            "\t\"animation_url\": \"", animationURL, "\",\n",
            "\t\"attributes\": [\n", attributes, "\t]\n",
            "}"
        );
    }

    function generateImgUrl(bool isSpecialVariant) internal view returns (string memory) {
        return string(abi.encodePacked(baseURI, isSpecialVariant ? specialVariantURI : defaultURI, imageExtension));
    }

    function generateAnimationUrl(bool isSpecialVariant) internal view returns (string memory) {
        return string(abi.encodePacked(baseURI, isSpecialVariant ? specialVariantURI : defaultURI, animationExtension));
    }

    function generateAttributes(ForgeToken.TokenAttributes memory tokenData, string memory tokenId) internal view returns (string memory) {
        return string(
            abi.encodePacked(
                generateSpecialVariantAttribute(tokenData),
                generateXpAttribute(tokenData),
                generateMintNumberAttribute(tokenId)
            )
        );
    }

    function generateSpecialVariantAttribute(ForgeToken.TokenAttributes memory tokenData) internal view returns (string memory) {
        return string(
            abi.encodePacked(
                "\t\t{\n",
                "\t\t\t\"trait_type\": \"", specialVariantTraitName, "\",\n",
                "\t\t\t\"value\":\"", tokenData.isSpecialVariant ? specialVariantTraitValue : defaultTraitValue, "\"\n",
                "\t\t},"
            )
        );
    }

    function generateXpAttribute(ForgeToken.TokenAttributes memory tokenData) internal view returns (string memory) {
        return string(
            abi.encodePacked(
                "\t\t{\n",
                "\t\t\t\"display_type\": \"", xpDisplayType, "\",\n",
                "\t\t\t\"trait_type\": \"", xpTraitName, "\",\n",
                "\t\t\t\"max_value\": 32760,\n",
                "\t\t\t\"value\":", Strings.toString(tokenData.xp), "\n",
                "\t\t},"
            )
        );
    }

    function generateMintNumberAttribute(string memory tokenId) internal view returns (string memory) {
        return string(
            abi.encodePacked(
                "\t\t{\n",
                "\t\t\t\"display_type\": \"", mintNumberDisplayType, "\",\n",
                "\t\t\t\"trait_type\": \"", mintNumberTraitName, "\",\n",
                "\t\t\t\"value\":", tokenId, "\n",
                "\t\t}\n"
            )
        );
    }
}